

Complex Number Addition and the Complex (Argand) Plane, Activity 3

■ Learning Goals:

- 1) Review mathematical and *Mathematica* content of the two preceding activities in order to solidify understanding and skill.
- 2) Learn how to use *Mathematica*'s [Manipulate](#) function to make interactive (“dynamic”) output that further solidifies understanding and skill.
- 3) Gain understanding about the algebraic and geometric meanings of complex number subtraction.

■ Prerequisites:

- 1) Familiarity with vectors: in physics, linear algebra, multivariable calculus, and/or differential equations
- 2) Comfort with abstract algebra, though not necessarily at the level of a course in abstract algebra
- 3) *Mathematica* content from Activities 1 and 2

■ Introduction:

To have a [deep understanding](#) of even a simple function, like the [squaring](#) function $w = f(z) = z^2$, it is good, but not sufficient, to just be familiar with its formula and how to use it to compute values. Understanding its “behavior” by the nature of its [graph](#) (when z is real), its [mapping properties](#) (when z is complex), and its [dynamic](#) properties under [iteration](#) — including properties that may be [unexpected](#) — is essential to gain more complete knowledge. Likewise, static pictures that illustrate mathematical concepts online and in books are great, but dynamic and interactive pictures, created on computers, are better. One of the main goals of this activity is to learn how to use the [Manipulate](#) function in *Mathematica* to solidify what we have learned about the geometry of complex numbers and their sums in Activities 1 and 2 of this learning module and to gain new knowledge about the geometry of complex number subtraction. The fact that [subtraction is defined in terms of addition](#) is the algebraic key that unlocks the

geometric knowledge.

■ Content:

Recall from Activity 2 that the geometric meaning of the formula for the sum of two complex numbers, $(a + bi) + (c + di) = (a + c) + (b + d)i$, is the same as that for the sum of the two-dimensional vectors $\mathbf{v} = \langle a, b \rangle$ and $\mathbf{w} = \langle c, d \rangle$. In fact, we can think of the complex numbers as actually being vectors and use the [parallelogram law](#) to add them. The following code generates a picture to illustrate this. You should review Activity 2 to understand how this code works.

```
Show[Graphics[{Thick, Red, Arrow[{0, 0}, {1, 3}], Arrow[{2, 2}, {3, 5}], Blue,
  Arrow[{0, 0}, {2, 2}], Arrow[{1, 3}, {3, 5}], Black, Arrow[{0, 0}, {3, 5}]}],
Graphics[Text[Style["v", Blue, Large], {1.3, 1}],
Graphics[Text[Style["v", Blue, Large], {1.6, 4}],
  Text[Style["w", Red, Large], {.4, 2}], Text[Style["w", Red, Large], {2.7, 3.2}],
  Rotate[Text[Style["v+w", Large], {1.3, 2.5}], 60 Degree]]],
Axes → True, PlotRange → {{-.1, 5.1}, {-.1, 5.1}}]
```

Discussion 1: Review the meaning of the [parallelogram law](#), then describe its meaning in your own words without looking at other written sources.

Response 1:

(You can type your thoughts and answers here formatted in text mode)

Grader/Instructor Response 1:

(The grader/instructor will give you feedback about your work here)

One of our goals here in Activity 3 is to learn how to use *Mathematica*'s key dynamic interactivity command, [Manipulate](#), to make slider-enabled animations that illustrate the parallelogram rule. First, we illustrate the use of [Manipulate](#) to make an interactive plot of the family of functions $f_\omega(x) = \sin(\omega x)$. Note that the [Plot](#) function is embedded within the [Manipulate](#), as the [Manipulate](#) function's first input (specifically, `Plot[Sin[ω*x], {x, -2π, 2π}, PlotRange → {-1, 1}, PlotStyle → {Thick, Cyan}, AxesLabel → {"x", "y"}]` is the first input of [Manipulate](#) below). The parameter ω is called the [angular frequency](#) of the sinusoidal function, and it also serves as our animation parameter. The list $\{\omega, 1, 5\}$ is the second (and last) input for [Manipulate](#) in this case, and it causes the value of ω to start at 1 and increase to 5 before starting over again when the animation is played. For a sampling of values of ω between 1 and 5, we get pictures of the graphs of the equations $y = f_\omega(x) = \sin(\omega x)$ that are stitched together to form an animation as ω increases.

```
Manipulate[Plot[Sin[ω * x], {x, -2 π, 2 π}, PlotRange → {-1, 1},
  PlotStyle → {Thick, Cyan}, AxesLabel → {"x", "y"}], {ω, 1, 5}]
```

As another example to illustrate the power and flexibility of [Manipulate](#), here is a short bit of code that will create a dynamic version of [Pascal's Triangle](#). Note here that [Binomial](#) computes the value of the [Binomial coefficient](#) $\binom{n}{k} = \frac{n!}{k!(n-k)!}$, [Table](#) creates the array of values, while [Column](#) and [Center](#) are used to format the output in the traditional way that Pascal's Triangle is displayed, though the alignment is poor near the left and right edges as the triangle increases in size. The use of the last 1 in the list $\{m, 0, 15, 1\}$ is what forces the value of m to increment up by 1 each time to remain at integer values as the animation proceeds.

```
Manipulate[Column[Table[Binomial[n, k], {n, 0, m}, {k, 0, n}], Center], {m, 0, 15, 1}]
```

Now we use [Manipulate](#) in conjunction with [Graphics](#) and [Arrow](#) to create an interactive output that illustrates the [parallelogram law](#) as the complex numbers change. In the code below, there are four animation parameters for [Manipulate](#): a , b , c , and d ; representing the real and imaginary parts of the complex numbers $a + bi$ and $c + di$. Take note of the effects of the options [ImageSize](#) and [LabelStyle](#) on the output.

```
Manipulate[
  Graphics[{Thick, Red, Arrow[{0, 0}, {a, b}], Arrow[{c, d}, {a + c, b + d}],
    Blue, Arrow[{0, 0}, {c, d}], Arrow[{a, b}, {a + c, b + d}],
    Black, Arrow[{0, 0}, {a + c, b + d}]}], Axes → True, AxesLabel →
  {Text[Style["real", Large, Italic]], Text[Style["imaginary", Large, Italic]]},
  TicksStyle → Large, PlotRange → {{-8.1, 8.1}, {-8.1, 8.1}}, ImageSize → Large],
  {{a, 1}, -4, 4}, {{b, 3}, -4, 4}, {{c, 2}, -4, 4},
  {{d, 2}, -4, 4}, LabelStyle → Large]
```

If we are careful, we can now use [Show](#), [Graphics](#), and [Text](#) to add textual labels to these complex numbers (vectors) that move appropriately as the numbers themselves move. We will use z_1 and z_2 as the labels for these complex numbers (vectors). Make sure you take the time to think about how the locations of the labels (as rectangular coordinates) within the [Text](#) commands (as, for example, $.5\{c,d\} + \{a,b\}$) depend on the values of a , b , c , and d . Note that the syntax near the end of the cell (such as the $\{b,3\}, -4, 4$) gives the “starting values” of a , b , c , and d as 1, 3, 2, and 2, respectively; and note that this syntax forces the animation parameters to vary over the range from -4 to 4 .

```
Manipulate[
  Show[Graphics[{Thick, Red, Arrow[{0, 0}, {a, b}], Arrow[{c, d}, {a + c, b + d}],
    Blue, Arrow[{0, 0}, {c, d}], Arrow[{a, b}, {a + c, b + d}],
    Black, Arrow[{0, 0}, {a + c, b + d}]}],
  Graphics[{Text[Style["z2", Blue, Large], .5 * {c, d}],
    Text[Style["z2", Blue, Large], .5 {c, d} + {a, b}], Text[Style["z1", Red, Large],
    .5 * {a, b}], Text[Style["z1", Red, Large], .5 {a, b} + {c, d}],
    Text[Style["z1+z2", Large], .5 {a + c, b + d}]}], Axes → True, AxesLabel →
  {Text[Style["real", Large, Italic]], Text[Style["imaginary", Large, Italic]]},
  TicksStyle → Large, PlotRange → {{-8.1, 8.1}, {-8.1, 8.1}}, ImageSize → Large],
  {{a, 1}, -4, 4}, {{b, 3}, -4, 4}, {{c, 2}, -4, 4},
  {{d, 2}, -4, 4}, LabelStyle → Large]
```

To create a *cursor-enabled*, rather than a slider-enabled, animation of the same picture, we can make use of the [Locator](#) command within [Manipulate](#). Note that the syntax gives the “starting values” of z_1 and z_2 as $z_1 = 1 + 3i$ and $z_2 = 2 + 2i$ (or $\{1, 3\}$ and $\{2, 2\}$ as lists). [Locator](#) then allows these to change two-dimensionally with the cursor location (once either “cross-hair” is clicked on) over the given window.

```
Manipulate[
  Show[Graphics[{Thick, Red, Arrow[{0, 0}, z1], Arrow[{z2, z1 + z2}], Blue, Arrow[
    {0, 0}, z2], Arrow[{z1, z1 + z2}], Black, Arrow[{0, 0}, z1 + z2]}], Graphics[
    {Text[Style["z2", Blue, Large], .5 * z2], Text[Style["z2", Blue, Large], .5 z2 + z1],
    Text[Style["z1", Red, Large], .5 * z1], Text[Style["z1", Red, Large], .5 * z1 + z2],
    Text[Style["z1+z2", Large], .5 * (z1 + z2)]}, Axes → True, AxesLabel →
    {Text[Style["real", Large, Italic]], Text[Style["imaginary", Large, Italic]]},
    TicksStyle → Large, PlotRange → {{-8.1, 8.1}, {-8.1, 8.1}}, ImageSize → Large],
  {{z1, {1, 3}}, Locator}, {{z2, {2, 2}}, Locator},
  LabelStyle → Large]
```

Mathematica Exercise I: Return to Activity 2 and study the way the sum of *three* complex numbers (as vectors) can be visualized. Next, use [Manipulate](#),

[Graphics](#) , [Arrow](#), [Show](#), [Text](#), and other *Mathematica* commands to create an interactive and labeled version of this diagram whose animation parameters are the real and imaginary parts of the complex numbers $z_1 = a + b i$, $z_2 = c + d i$, and $z_3 = e + f i$. You can choose to make your interactivity either slider-enabled or cursor-enabled.

Mathematica Work 1:

(Enter your code under this cell when *Mathematica* is in “Input mode” — make sure a horizontal line is showing before you start typing)

Grader/Instructor Mathematica Assessment 1:

(The grader/instructor will give you feedback about your work here)

The operation of subtraction can be defined for any algebraic structure where addition has been defined in such a way so as to guarantee the existence of an additive identity and additive inverses. In such a situation, one element can then be subtracted from another element by adding the additive inverse of the element being subtracted. In the particular case of the set of complex numbers \mathbb{C} , once addition has been defined via the formula $(a + b i) + (c + d i) = (a + c) + (b + d) i$, it is clear that $0 + 0 i$ is the additive identity and that $(-a) + (-b) i$ is the additive inverse of the arbitrary complex number $a + b i$. Hence, we can define $(a + b i) - (c + d i)$ to equal $(a + b i) + ((-c) + (-d) i)$. Clearly this is also the same as $(a - c) + (b - d) i$, which is what you hopefully would have guessed we should do for complex number subtraction before reading this paragraph.

The geometric meaning of complex number subtraction is the same as the geometric meaning of vector subtraction in the plane. Given two vectors $\mathbf{v} = \langle a, b \rangle$ and $\mathbf{w} = \langle c, d \rangle$, how should we geometrically interpret the meaning of $\mathbf{v} - \mathbf{w}$? There are two typical ways of doing this, both related to the parallelogram law for vector addition. We could interpret $\mathbf{v} - \mathbf{w}$ as $\mathbf{v} + (-\mathbf{w})$, where $-\mathbf{w} = \langle -c, -d \rangle$ is the additive inverse of $\mathbf{w} = \langle c, d \rangle$, and $-\mathbf{w}$ has the same length as \mathbf{w} but points in the exact opposite direction. Or we could realize, knowing that vector addition is commutative, that $\mathbf{v} - \mathbf{w}$ is the unique vector with the property that it gives \mathbf{v} when added to \mathbf{w} ; that is, $\mathbf{w} + (\mathbf{v} - \mathbf{w}) = (\mathbf{v} - \mathbf{w}) + \mathbf{w} = \mathbf{v}$. If we view complex numbers as two-dimensional vectors, then we should view the geometry of complex number subtraction in the same way. Both of these perspectives are represented in the interactive diagram generated by the code below for the complex numbers $z_1 = a + b i$ and $z_2 = c + d i$ and their difference

$$z_1 - z_2 = (a - c) + (b - d) i.$$

```
Manipulate[
  Show[Graphics[{Thick, Red, Arrow[{0, 0}, z1]}, LightRed, Arrow[{-z2, z1 - z2}],
    Blue, Arrow[{0, 0}, z2]}, Arrow[{z1 - z2, z1}], LightBlue,
    Arrow[{0, 0}, -z2]}, Green, Arrow[{z2, z1}], Arrow[{0, 0}, z1 - z2]}],
  Graphics[{Text[Style["z2", Blue, Large], .5 * z2], Text[Style["z2", Blue, Large],
    .5 z2 + (z1 - z2)], Text[Style["-z2", LightBlue, Large], -.5 * z2],
    Text[Style["z1", Red, Large], .5 * z1], Text[Style["z1", LightRed, Large],
    .5 * z1 - z2], Text[Style["z1-z2", Green, Large], .5 * (z1 + z2)],
    Text[Style["z1-z2", Green, Large], .5 * (z1 - z2)]}, Axes -> True, AxesLabel ->
  {Text[Style["real", Large, Italic]], Text[Style["imaginary", Large, Italic]]},
  TicksStyle -> Large, PlotRange -> {{-8.1, 8.1}, {-8.1, 8.1}}, ImageSize -> Large],
  {{z1, {1, 3}}, Locator}, {{z2, {2, 2}}, Locator},
  LabelStyle -> Large]
```

Because of this diagram, the “quickest” way to draw $z_1 - z_2$ as a vector is to draw a vector based at the

tip (head) of z_2 and terminating at the tip (head) of z_1 , when both z_1 and z_2 are drawn starting from the same base (which is typically the origin $0 = 0 + 0i$).

Discussion 2: Give a detailed explanation of why the diagram created from the preceding line of code illustrates the geometric meaning of complex number subtraction in two different ways, based on the [parallelogram law](#) for complex number addition. Note that $-z_2$ can be thought of as playing a role in this diagram in two locations on one the parallelograms, though it is only labeled as $-z_2$ once.

Response 2:

(You can type your thoughts and answers here formatted in text mode)

Grader/Instructor Response 2:

(The grader/instructor will give you feedback about your work here)

Mathematica Exercise 2: Use [Manipulate](#), [Graphics](#), [Arrow](#), [Show](#), [Text](#), and other *Mathematica* commands to create an interactive diagram to illustrate the geometric meaning of $z_1 - z_2 - z_3 = z_1 + (-z_2) + (-z_3) = z_1 - (z_2 + z_3)$ in any way you like. You can choose to make your interactivity either slider-enabled or cursor-enabled.

Mathematica Work 2:

(Enter your code under this cell when *Mathematica* is in “Input mode” — make sure a horizontal line is showing before you start typing)

Grader/Instructor *Mathematica* Assessment 2:

(The grader/instructor will give you feedback about your work here)

In our final activity, Activity 4, of this learning module on complex addition and the complex plane, we will delve a bit into [complex number multiplication](#) and the idea of [complex conjugation](#) for the purpose of understanding a fundamental inequality called the [triangle inequality](#), which symbolically expresses the basic idea that the shortest distance between two points is along the straight line between them.